
Abstract. For Markov regime-switching models, a nonstandard test statistic
must be used to test for the possible presence of multiple regimes. Carter and
Steigerwald (2013, Journal of Econometric Methods 2: 25–34) derive the ana-
lytic steps needed to implement the Markov regime-switching test proposed by
Cho and White (2007, Econometrica 75: 1671–1720). We summarize the imple-
mentation steps and address the computational issues that arise. We then in-
troduce a new command to compute regime-switching critical values, rscv, and
present it in the context of empirical research.

Keywords: st0347, rscv, Markov regime switching

1 Introduction

Markov regime-switching models are frequently used in economic analysis and are preva-
lent in fields such as finance, industrial organization, and business cycle theory. Unfortu-
nately, conducting proper inference with these models can be exceptionally challenging.
In particular, testing for the possible presence of multiple regimes requires the use of a
nonstandard test statistic and critical values that may differ across model specifications.

Cho and White (2007) demonstrate that because of the unusually complicated na-
ture of the null space, the appropriate measure for a test of multiple regimes in the
Markov regime-switching framework is a quasi-likelihood-ratio (QLR) statistic. They
provide an asymptotic null distribution for this test statistic from which critical values
should be drawn. Because this distribution is a function of a Gaussian process, the
critical values are difficult to obtain from a simple closed-form distribution. Moreover,
the elements of the Gaussian process underlying the asymptotic null distribution are
dependent upon one another. Thus the critical values depend on the covariance of the
Gaussian process and, because of the complex nature of this covariance structure, are
best calculated using numerical approximation. In this article, we summarize the steps
necessary for such an approximation and introduce the new command rscv, which can
be used to produce the desired regime-switching critical values for a QLR test of only
one regime.

We focus on a simple linear model with Gaussian errors, but the QLR test and the
rscv command are generalizable to a much broader class of models. This methodology
can be applied to models with multiple covariates and non-Gaussian errors. It is also

c© 2014 StataCorp LP st0347



applicable to regime-switching models where the dependent variable is vector valued,
although the difference between distributions must be in only one mean parameter.
Although most regime-switching models are thought of in the context of time-series data,
we provide an example in section 5 of how to use the QLR test in cross-section models.
However, there is one notable restriction on the allowable class of regime-switching
models. Carter and Steigerwald (2012) establish that the quasi-maximum likelihood
estimator created using the quasi-log-likelihood is inconsistent if the covariates include
lagged values of the dependent variable. Thus the QLR test should be used with extreme
caution on autoregressive models.

The article is organized as follows. In section 2, we describe the unusual null space
that corresponds to a test of only one regime versus the alternative of regime switching.
In section 3, we present the QLR test statistic, as derived by Cho and White (2007),
and the corresponding asymptotic null distribution. We also summarize the analysis
in Carter and Steigerwald (2013) describing the covariance structure of the relevant
Gaussian process. In section 4, we describe the methodology used by the rscv command
to numerically approximate the relevant critical values. We also present the syntax and
options of the rscv command and provide sample output. We illustrate the use of the
rscv command with an application from the economics literature in section 5. Finally,
we conclude in section 6 with some remarks on the general applicability of this command
and the underlying methods.

2 Null hypothesis

Specifying a Markov regime-switching model requires a test to confirm the presence of
multiple regimes. The first step is to test the null hypothesis of one regime against the
alternative hypothesis of Markov switching between two regimes. If this null hypothesis
can be rejected, then one can proceed to estimate the Markov regime-switching models
with two or more regimes. The key to conducting valid inference is then a test of the
null hypothesis of one regime, which yields an asymptotic size equal to or less than the
nominal test size.

To understand how to conduct valid inference for the null hypothesis of only one
regime, consider a basic regime-switching model,

yt = θ0 + δst + ut (1)

where ut ∼ i.i.d.N
(
0, σ2

)
. The unobserved state variable st ∈ (0, 1) indicates that

regime in state 0, yt has mean θ0, while regime in state 1, yt has mean θ1 = θ0+ δ. The
sequence (st)

n
t=1 is generated by a first-order Markov process with P (st = 1|st−1 = 0) =

p0 and P (st = 0|st−1 = 1) = p1.

The key is to understand the parameter space that corresponds to the null hypoth-
esis. Under the null hypothesis, there is one regime with mean θ∗. Hence, the null
parameter space must capture all the possible regions that correspond to one regime.
The first region corresponds to the assumption that θ0 = θ1 = θ∗, which is the as-
sumption that each of the two regimes is observed with positive probability: p0 > 0



and p1 > 0. The nonstandard feature of the null space is that it includes two addi-
tional regions, each of which also corresponds to one regime with mean θ∗. The second
region corresponds to the assumption that only regime 0 occurs with positive probabil-
ity, p0 = 0, and that θ0 = θ∗. In this second region, the mean of regime 1, θ1 is not
identified, so this region in the null hypothesis does not impose any value on θ1 − θ0.
The third region is a mirror image of the second region, where now the assumption is
that regime 1 occurs with probability 1: p1 = 0 and θ1 = θ∗. The three regions are
depicted in figure 1. The vertical distance measures the value of p0 and of p1, and the
horizontal distance measures the value of θ1 − θ0. Thus the vertical line at θ1 = θ0
captures the region of the null parameter space that corresponds to the assumption
that θ0 = θ1 = θ∗ together with p0, p1 ∈ (0, 1). The lower horizontal line captures the
region of the null parameter space where p0 = 0 and θ1 − θ0 is unrestricted. Similarly,
the upper horizontal line captures the region of the null parameter space where p1 = 0
and θ1 − θ0 is unrestricted.

θ1 − θ0 = 0
p1 = 0

p0 = 0

Figure 1. All three regions of the null hypothesis H0 : p0 = 0 and θ0 = θ∗; p1 = 0 and
θ1 = θ∗; or θ0 = θ1 = θ∗ together with local neighborhoods of p1 = 0 and θ0 = θ1 = θ∗

The additional curves that correspond to the values p0 = 0 and p1 = 0 help prevent
one from misclassifying a small group of extremal values as a second regime. In figure 1,
we depict the null space together with local neighborhoods for two points in this space.
These two neighborhoods illustrate the different roles of the three curves in the null
space. Points in the circular neighborhood of the point on θ1 − θ0 = 0 correspond
to processes with two regimes that have only slightly separated means. Points in the
semicircular neighborhood around the point on p1 = 0 correspond to processes in which
there are two regimes with widely separated means, one of which occurs infrequently.
Because a researcher is often concerned that rejection of the null hypothesis of one
regime is due to a small group of outliers rather than multiple regimes, including these
boundary values reduces this type of false rejection. Consequently, a valid test of the
null hypothesis of one regime must account for the entire null region and include all
three curves.



3 QLR test statistic

To implement a valid test of the null hypothesis of one regime, a likelihood-ratio statistic
is needed. When considering the likelihood-ratio statistic for a Markov regime-switching
process, Cho and White (2007) find that including p0 = 0 and p1 = 0 in the parameter
space creates significant difficulties in the asymptotic analysis. These difficulties lead
them to consider a QLR statistic for which the Markov structure of the state variable is
ignored and (st) is instead a sequence of independent and identically distributed (i.i.d.)
random variables.

This i.i.d. restriction allows Cho and White (2007) to consider only the stationary
probability, P (st = 1) = π, where π = p0/(p0+ p1). Because π = 1 if and only if p1 = 0
(and π = 0 if and only if p0 = 0), the null hypothesis for a test of one regime based on
the QLR statistic is expressed with three curves. The null hypothesis is H0 : θ0 = θ1 = θ∗
(curve 1), π = 0 and θ0 = θ∗ (curve 2), and π = 1 and θ1 = θ∗ (curve 3). The alternative
hypothesis is H1 : π ∈ (0, 1) and θ0 6= θ1.

For our basic model in (1), the quasi-log-likelihood analyzed by Cho and White
(2007) is

Ln

(
π, σ2, θ0, θ1

)
=

1

n

n∑

t=1

lt
(
π, σ2, θ0, θ1

)

where lt(π, σ
2, θ0, θ1) := log{(1 − π)f(yt|σ2, θ0) + πf(yt|σ2, θ1)} and f(yt|σ2, θj) is the

conditional density with j = 0, 1. (π̂, σ̂2, θ̂0, θ̂1) are the parameter values that maximize

the quasi-log-likelihood function. (1, σ̃2, ·, θ̃1) are the parameter values that maximize
Ln under the null hypothesis that π = 1. The QLR statistic is then

QLRn = 2n
{
Ln

(
π̂, σ̂2, θ̂0, θ̂1

)
− Ln

(
1, σ̃2, ·, θ̃1

)}

The asymptotic null distribution of QLRn is (Cho and White 2007, theorem 6(b),
1692),

QLRn ⇒ max

[
{max (0, G)}2 , sup

Θ

{
G (θ0)−

}2
]

(2)

where G(θ0) is a Gaussian process, G(θ0)− := min{0,G(θ0)}, and G is a standard Gaus-
sian random variable correlated with G(θ0). (For a more complete description of (2),
see Bostwick and Steigerwald [2012]).

The critical value for a test based on the statistic QLRn thus corresponds to a quan-
tile for the largest value over max(0, G)2 and supΘ{G(θ0)−}2. To determine this quan-
tity, one must account for the covariance among the elements of G(θ0) as well as their
covariance with G. The structure of this covariance, which is described in detail in
Bostwick and Steigerwald (2012), is

E {G (θ0)G (θ′0)} =
eηη

′ − 1− ηη′ − (ηη′)
2

2(
eη2 − 1− η2 − η4

2

) 1

2

{
e(η′)2 − 1− (η′)

2 − (η′)4

2

} 1

2

(3)



where η = (θ0 − θ∗)/σ and η′ = (θ′0 − θ∗)/σ. This covariance determines the quantity
supΘ{G(θ0)−}2 that appears in the asymptotic null distribution. Because the regime-
specific parameters enter (3) only through η, a researcher does not need to specify the
parameter space Θ to calculate supΘ{G(θ0)−}2. The only requirement is to specify the
set H that contains the number of standard deviations that separate the regime means.
Finally, to fully capture the behavior of the asymptotic null distribution of QLRn, we
must also account for the covariance between G and G(θ0). Cho and White (2007) show

that Cov{G,G(θ0)} = (eη
2 − 1− η2 − η4/2)−1/2η4.

4 The rscv command

4.1 Syntax

rscv
[
, ll(#) ul(#) r(#) q(#)

]

4.2 Description

rscv simulates the asymptotic null distribution of QLRn and returns the corresponding
critical value. If no options are specified, rscv returns the critical value for a size 5%
QLR test with a regime separation of ±1 standard deviation calculated over 100,000
replications.

4.3 Options

ll(#) specifies a lower bound on the interval H containing the number of standard
deviations separating regime means, where η ∈ H. The default is ll(-1), meaning
that the mean of regime 1 is no more than 1 standard deviation below the mean of
regime 2.

ul(#) specifies an upper bound on the interval H containing the number of standard
deviations separating regime means. The default is ul(1), meaning that the mean
of regime 1 is no more than 1 standard deviation above the mean of regime 2.

r(#) specifies the number of simulation replications to be used in calculating the critical
values. The default is r(100000), meaning that the simulation will be run 100,000
times.

q(#) specifies the quantile for which a critical value should be calculated. The default
is q(0.95), which corresponds to a nominal test size of 5%.

4.4 Simulation process

For a QLR test with size 5%, the critical value corresponds to the 0.95 quantile of the
limit distribution given on the right side of (2). Because the dependence in the process



G (θ0) renders numeric integration infeasible, we construct the quantile by simulating
independent replications of the process. In this section, we describe the simulation
process used to obtain these critical values and how each of the rscv command options
affects those simulations.

Because the covariance of G (θ0) depends on only an index η, we do not need to
simulate G (θ0) directly. Instead, we simulate GA (η), which we will construct to have
the same covariance structure as G (θ0). The process GA (η) will therefore provide us
with the correct quantile while relying solely on the index, η.

To construct GA (η) for the covariance structure in (3), recall that by a Taylor-series
expansion, eη = 1 + η + η2/2! + · · · . Hence, for (ǫk)∞k=0 ∼ i.i.d.N (0, 1),

∞∑

k=3

ηk√
k!
ǫk ∼ N

(
0, eη

2 − 1− η2 − η4

2

)

Using this fact, our simulated process is constructed as

GA (η) =

(
eη

2 − 1− η2 − η4

2

)− 1

2
K−1∑

k=3

ηk√
k!
ǫk

where K determines the accuracy of the Taylor-series approximation. Note that the
covariance of this simulated process, E{GA(η)GA(η′)}, is identical to the covariance
structure of G(θ0) in (3).

We must also account for the covariance between G and G(θ0). Cho and White
(2007) establish that this covariance corresponds to the term in the Taylor-series ex-
pansion for k = 4. Thus we set G = ǫ4 so that Cov{G,G(θ0)} = Cov{G,GA(η)}.
Therefore, the critical value that corresponds to (2) for a test size of 5% is the 0.95
quantile of the simulated value

max

(
{max (0, ǫ4)}2 ,max

η∈H

[
min

{
0,GA (η)

}]2
)

(4)

The rscv command executes the numerical simulation of (4) by first generating the
series (ǫk)

K
k=0 ∼ i.i.d.N(0, 1). For each value in a discrete set of η ∈ H, it then constructs

GA(η) = (eη
2 − 1 − η2 − η4/2)−1/2

∑K−1
k=3 ηk/

√
k!ǫk. The command then obtains the

value mi = max({max(0, ǫ4)}2,maxη[min{0,GA(η)}]2), corresponding to (2) for each
replication (indexed by i). Let (m[i])

r
i=1 be the vector of ordered values of mi calculated

in each replication. The command rscv returns the critical value for a test with size q
from m[(1−q)r].

For each replication, rscv calculates GA(η) at a fine grid of values over the interval
H. To do so requires three quantities: the interval H (which must encompass the true
value of η), the grid of values over H (given by the grid mesh), and the number of
desired terms in the Taylor-series approximation, K. The user specifies the interval H
using the ll() and ul() options. If θ0 is thought to lie within 3 standard deviations



of θ1, the interval is H = [−3.0, 3.0]. Because the process is calculated at only a finite
number of values, the accuracy of the calculated maximum increases as the grid mesh
shrinks. Thus the command rscv implements a grid mesh of 0.01, as recommended in
Cho and White (2007, 1693). For the interval H = [−3.0, 3.0], and with a grid mesh of
0.01, the process is calculated at the points (−3.00,−2.99, . . . , 3.00).

Given the grid mesh of 0.01 and the user-specified interval H, we must determine
the appropriate value of K. To do so, we consider the approximation error, ξK,η =

(eη
2 − 1 − η2 − η4/2)−1/2

∑∞
k=K ηk/

√
k!ǫk. We want to ensure that as K increases,

the variance of ξK,η decreases toward zero. Carter and Steigerwald (2013) show that
for large K, var(ξK,η) ≤ e2J log η−K logK . Therefore, the command rscv implements a
value of K such that for the user-specified interval H, (maxH |η|)2/K ≤ 1/2.

The rscv command also allows the user to specify the number of simulation repli-
cations and the desired quantile. For large values of H and the default number of
replications (r = 100000), the rscv command could require more memory than a 32-bit
operating system can provide. In this case, the user may need to specify a smaller num-
ber of replications to calculate the critical values for the desired interval, H. Critical
values derived using fewer simulation replications may be stable to only one significant
digit. Table 1 depicts the results of rscv for a size 5% test over varying values of ll(),
ul(), and r().

Table 1. Critical values for linear models with Gaussian errors

H (−1, 1) (−2, 2) (−3, 3) (−4, 4) (−5, 5)

100,000 4.9 5.6 6.2 6.7 7.0
Replications

10,000 4.9 5.6 6.2 6.6 7.1

Nominal level 5%; grid mesh of 0.01.

5 Example

We demonstrate how to test for the presence of multiple regimes through an example
from the economics literature. Unlike the simple model that we have considered until
now, (1), the model in this example includes several added complexities that are com-
monly used in regime-switching applications. We describe how to construct the QLR

test statistic for this more general model, how to use existing Stata commands to obtain
the value of the test statistic, and, finally, how to use the new command, rscv, to obtain
an appropriate critical value.

Our example is derived from Bloom, Canning, and Sevilla (2003), who test whether
the large differences in income levels across countries are better explained by differences
in intrinsic geography or by a regime-switching model where the regimes correspond to



distinct equilibria. To this end, the authors use cross-sectional data to analyze the dis-
tribution of per capita income levels for countries with similar exogenous characteristics
and test for the presence of multiple regimes.

Bloom, Canning, and Sevilla (2003) propose a model of switching between two pos-
sible equilibria. Regime 1 occurs with probability p(x) and corresponds to countries
that are in a poverty trap equilibrium.

y = µ1 + β1x+ ǫ1 , Var(ǫ1) = σ2
1 (5)

Regime 2 occurs with probability 1 − p(x) and corresponds to countries in a wealthy
equilibrium.

y = µ2 + β2x+ ǫ2 , Var(ǫ2) = σ2
2 (6)

In both regimes, y is the log gross domestic product per capita, and x is the absolute lat-
itude, which functions as a catchall for a variety of exogenous geographic characteristics.
This model differs from a Markov regime-switching model in that the authors are look-
ing at different regimes in a cross-section rather than over time. Thus the probability of
being in either regime is stationary, and the unobserved regime indicator is an i.i.d. ran-
dom variable. This modification corresponds exactly to that made by Cho and White
(2007) to create the quasi-log-likelihood, so in this example, the log-likelihood ratio and
the QLR are one and the same.

Note that this model is more general than the basic regime-switching model pre-
sented in section 2. Bloom, Canning, and Sevilla (2003) have allowed for three general-
izations: covariates with coefficients that vary across regimes; error variances that are
regime specific; and regime probabilities that depend on the included covariates. How-
ever, as Carter and Steigerwald (2013) discuss, the asymptotic null distribution (2) is
derived under the following assumptions: that the difference between regimes be in only
the intercept µj ; that the variance of the error terms be constant across regimes; and
that the regime probabilities do not depend on the exogenous characteristic, x. Thus,
to form the test statistic, we must fit the following two-regime model: regime 1 occurs
with probability p and corresponds to

y = µ1 + βx+ ǫ (5′)

while regime 2, which occurs with probability (1− p), corresponds to

y = µ2 + βx+ ǫ (6′)

where Var (ǫ) = σ2.

Simplifying the model like this does not diminish the validity of the QLR as a one-
regime test for the model in (5) and (6). Under the null hypothesis of one regime, there is
necessarily only one error variance, only one coefficient for each covariate, and a regime
probability equal to one. Thus, under the null hypothesis, the QLR test will necessarily
have the correct size even if the data are accurately modeled by a more complex system.



Once the null hypothesis is rejected using this restricted model, the researcher can then
fit a model with regime-specific variances and coefficients, if desired.1

For the restricted model in (5′) and (6′), the quasi-log-likelihood is

Ln

(
p, σ2, β, µ1, µ2

)
=

1

n

n∑

t=1

lt
(
p, σ2, β, µ1, µ2

)

where lt(p, σ
2, β, µ1, µ2) := log{pf(yt|xt;σ

2, β, µ1) + (1− p)f(yt|xt;σ
2, β, µ2)}, and

f(yt|xt;σ
2, β, µj) is the conditional density for j = 1, 2. It is common to assume, as

Bloom, Canning, and Sevilla (2003) do, that ǫ is a normal random variable2 so that

f(yt|xt;σ
2, β, µj) = 1/(

√
2πσ2)e−(yt−µj−βxt)

2/(2σ2). Let (p̂, σ̂2, β̂, µ̂1, µ̂2) be the values

that maximize Ln and let (1, σ̃2, β̃, µ̃1, ·) be the values that make Ln as large as possible
under the null hypothesis of one regime. The QLR statistic is then

QLRn = 2n
{
Ln

(
p̂, σ̂2, β̂, µ̂1, µ̂2

)
− Ln

(
1, σ̃2, β̃, µ̃1, ·

)}

To estimate QLRn, we use the same Penn World Table and CIA World Factbook data
as in Bloom, Canning, and Sevilla (2003).3 First, we must determine the parameter

values that maximize the quasi-log-likelihood under the null hypothesis, (1, σ̃2, β̃, µ̃1, ·)
and evaluate the quasi-log-likelihood at those values. To obtain these parameter values,
we estimate a linear regression of y on x, which corresponds to maximizing

Ln

(
1, σ2, β, µ1, ·

)
=

1

n

n∑

t=1

log

(
1√
2πσ2

e
−1

2σ2
(yt−µ1−βxt)

2

)

While this can be achieved with a simple ordinary least-squares command, we also need
the value of the log-likelihood, so we detail how to use Stata commands to obtain both
the parameter estimates and this value.

1. With a more complex data-generating process, these restrictions could lead to an increased prob-
ability of failing to reject a false null hypothesis and, hence, a decrease in the power of the QLR
test.

2. Bloom, Canning, and Sevilla (2003) assume normally distributed errors, but the QLR test allows
for any error distribution within the exponential family.

3. Latitude data for countries appearing in the 1985 Penn World Tables and missing from the CIA
World Factbook come from https://www.google.com/.



To find (1, σ̃2, β̃, µ̃1, ·), we use the following code, which relies on the Stata command
ml.

. program define llfsingle
1. version 13
2. args lnf mu beta sigma
3. quietly replace `lnf´= (1/_N)*ln(((2*_pi*`sigma´^2)^(-1/2))*

> exp((-1/(2*`sigma´^2))*(lgdp-`mu´-`beta´*latitude)^2))
4. end

. ml model lf llfsingle /mu /beta /sigma

. ml maximize

initial: log likelihood = -<inf> (could not be evaluated)
feasible: log likelihood = -127.9261
rescale: log likelihood = -31.297788
rescale eq: log likelihood = -2.3397622
Iteration 0: log likelihood = -2.3397622 (not concave)
Iteration 1: log likelihood = -1.5884033 (not concave)
Iteration 2: log likelihood = -1.2842957
Iteration 3: log likelihood = -1.2479471
Iteration 4: log likelihood = -1.1988284
Iteration 5: log likelihood = -1.1982503
Iteration 6: log likelihood = -1.1982487
Iteration 7: log likelihood = -1.1982487

Number of obs = 152
Wald chi2(0) = .

Log likelihood = -1.1982487 Prob > chi2 = .

Coef. Std. Err. z P>|z| [95% Conf. Interval]

mu
_cons 6.927805 1.420095 4.88 0.000 4.144469 9.711141

beta
_cons .0408554 .049703 0.82 0.411 -.0565607 .1382714

sigma
_cons .8019654 .5670752 1.41 0.157 -.3094815 1.913412

. matrix gammasingle=e(b)

Then, using these estimates, we evaluate Ln at its maximum to find Ln(1, σ̃
2, β̃, µ̃1, ·).

. generate llf1regime=ln(((2*_pi*gammasingle[1,3]^2)^(-1/2))*
> exp((-1/(2*gammasingle[1,3]^2))*
> (lgdp-gammasingle[1,1]-gammasingle[1,2]*latitude)^2))

. quietly summarize llf1regime

. quietly replace llf1regime=r(sum)

. display "Final estimated quasi-log-likelihood for one regime: " llf1regime
Final estimated quasi-log-likelihood for one regime: -182.1338

Thus we have n× Ln(1, σ̃
2, β̃, µ̃1, ·) = −182.1388.

Second, we must determine the parameter values that maximize the quasi-log-
likelihood under the alternative hypothesis of two regimes, (p̂, σ̂2, β̂, µ̂1, µ̂2) and evaluate
the quasi-log-likelihood at those values. Direct maximization is more difficult under the



alternative hypothesis, because the quasi-log-likelihood involves the log of the sum of
two terms.

Ln

(
p, σ2, β, µ1, µ2

)
=

1

n

n∑

t=1

log
{
pf

(
yt|xt;σ

2, β, µ1

)
+ (1− p) f

(
yt|xt;σ

2, β, µ2

)}

The expectations-maximization (EM) algorithm is a method used to circumvent this
difficulty. This algorithm requires iterative estimation of the latent regime probabilities,
p, and maximization of the resultant log-likelihood function until parameter estimates
converge. The EM algorithm proceeds as follows:

1. Choose starting guesses for the parameter values p(0), σ2(0), β(0), µ
(0)
1 , µ

(0)
2 .

2. For each observation, calculate ηt = P(st = 1|yt, xt) such that

η̂t = p(0)
f
(
yt|xt;σ

2(0), β(0), µ
(0)
1

)

p(0)f
(
yt|xt;σ2(0), β(0), µ

(0)
1

)
+

(
1− p(0)

)
f
(
yt|xt;σ2(0), β(0), µ

(0)
2

)

3. Use Stata’s ml command to find the parameter values p(1), σ2(1), β(1), µ
(1)
1 , µ

(1)
2

that maximize the complete log-likelihood.

LC
n

(
p, σ2, β, µ1, µ2

)
=

1

n

n∑

t=1

{
η̂t log f

(
yt|xt;σ

2, β, µ1

)

+ (1− η̂t) log f
(
yt|xt;σ

2, β, µ2

)

+ (1− η̂t) log(1− p) + η̂t log p }

4. To test for convergence, calculate

a. max
{(

p(1), σ2(1), β(1), µ
(1)
1 , µ

(1)
2

)
−
(
p(0), σ2(0), β(0), µ

(0)
1 , µ

(0)
2

)}
;

b. |LC
n

(
p(1), σ2(1), β(1), µ

(1)
1 , µ

(1)
2

)
− LC

n

(
p(0), σ2(0), β(0), µ

(0)
1 , µ

(0)
2

)
|; and

c. (using numeric derivatives) max(LC
n
′
).

5. If all 3 convergence criteria are less than some tolerance level (we use 1/n), then

quit and use p(1), σ2(1), β(1), µ
(1)
1 , µ

(1)
2 as the final parameter estimates. Otherwise,

repeat steps 2–5 with p(1), σ2(1), β(1), µ
(1)
1 , µ

(1)
2 as the new starting guesses.



The following code illustrates the implementation of these steps to obtain (p̂, σ̂2, β̂,
µ̂1, and µ̂2).

. program define llfmulti
1. version 13
2. args lnf mu1 mu2 beta sigma p
3. quietly replace `lnf´= (1/_N)*((1-etahat)*(ln((2*_pi*`sigma´^2)^(-1/2))+

> ((-1/(2*`sigma´^2))*(lgdp-`mu2´-`beta´*latitude)^2)+
> ln(1-`p´))+etahat*(ln((2*_pi*`sigma´^2)^(-1/2))+
> ((-1/(2*`sigma´^2))*(lgdp-`mu1´-`beta´*latitude)^2)+ln(`p´)))

4. end

. generate error=10

. generate tol=1/_N

. while error>tol {
2. quietly replace f1=((2*_pi*gammahat[1,4]^2)^(-1/2))*

> exp((-1/(2*gammahat[1,4]^2))*(lgdp-gammahat[1,1]-gammahat[1,3]*latitude)^2)
3. quietly replace f2=((2*_pi*gammahat[1,4]^2)^(-1/2))*

> exp((-1/(2*gammahat[1,4]^2))*(lgdp-gammahat[1,2]-gammahat[1,3]*latitude)^2)
4. quietly replace fboth=gammahat[1,5]*f1+(1-gammahat[1,5])*f2
5. quietly replace etahat=gammahat[1,5]*f1/fboth
6. ml model lf llfmulti /mu1 /mu2 /beta /sigma /p
7. ml init gammahat, copy
8. quietly ml maximize
9. matrix gammanew=e(b)
10. *Check for convergence using user-defined program nds

. nds
11. quietly replace error=max(nd1,nd2,nd3,nd4,nd5)
12. matrix gammahat=gammanew
13. }

. ml display

Number of obs = 152
Wald chi2(0) = .

Log likelihood = -1.4441013 Prob > chi2 = .

Coef. Std. Err. z P>|z| [95% Conf. Interval]

mu1
_cons 6.532847 1.148891 5.69 0.000 4.281062 8.784632

mu2
_cons 7.813265 1.45266 5.38 0.000 4.966102 10.66043

beta
_cons .0451607 .0374139 1.21 0.227 -.0281691 .1184905

sigma
_cons .5986278 .4232938 1.41 0.157 -.2310128 1.428268

p
_cons .7708245 .4203024 1.83 0.067 -.052953 1.594602



Using these estimates, we evaluate Ln at its maximum to find Ln(p̂, σ̂
2, β̂, µ̂1, µ̂2).

. quietly replace f1=((2*_pi*gammanew[1,4]^2)^(-1/2))*
> exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,1]-gammanew[1,3]*latitude)^2)

. quietly replace f2=((2*_pi*gammanew[1,4]^2)^(-1/2))*
> exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,2]-gammanew[1,3]*latitude)^2)

. generate lf2reg=gammanew[1,5]*f1+(1-gammanew[1,5])*f2

. generate llf2regime=ln(lf2reg)

. quietly summarize llf2regime

. quietly replace llf2regime=r(sum)

. display "Final estimated quasi-log-likelihood for two regimes: " llf2regime
Final estimated quasi-log-likelihood for two regimes: -179.9662

Thus we have n×Ln(p̂, σ̂
2, β̂, µ̂1, µ̂2) = −179.9662. Then, to calculate the test statistic,

QLRn, we type

. generate QLR=2*(llf2reg-llf1reg)

. display "Quasi-likelihood-ratio test statistic of one regime: " QLR
Quasi-likelihood-ratio test statistic of one regime: 4.3352051

These estimates and the resulting QLR test statistic are summarized in table 2. For the
complete Stata code used to create table 2, see the appendix.

Table 2. QLR test of one regime versus two regimes

One regime Two regimes

Regime I Regime II

Constant (µ1, µ2) 6.928 6.533 7.813
Latitude (β) 0.041 0.045
Standard deviation of error (σ) 0.802 0.599
Probability of regime I (p) 0.771
Log likelihood (Ln) −182.1 −180.0
QLRn 4.3

Finally, we use the rscv command to calculate the critical value for the QLR test of
size 5%. We allow for the possibility that the two regimes are widely separated and set
H = (−5.0, 5.0). The command and output are shown below.

. rscv, ll(-5) ul(5) r(100000) q(0.95)
7.051934397

Given that this critical value of 7.05 exceeds the QLR statistic of 4.3, we cannot reject
the null hypothesis of one regime.

This result is consistent with the findings of Bloom, Canning, and Sevilla (2003),
although they use a different method to obtain the necessary critical values. They



report a likelihood ratio and the corresponding critical values for a restricted version of
their model where the regime probabilities are fixed (p does not depend on x). Using
this restricted model, the authors do not reject the null hypothesis of one regime. At
the time that Bloom, Canning, and Sevilla (2003) were published, researchers had yet to
successfully derive the asymptotic null distribution for a likelihood-ratio test of regime
switching. Therefore, the authors use Monte Carlo methods to generate their critical
values using random data generated from the estimated relationship given by the model
in (5) and (6). The primary disadvantage of this approach is that the derived critical
values are then dependent upon the authors’ assumptions concerning the underlying
data-generating process.

Bloom, Canning, and Sevilla (2003) go on to report a likelihood-ratio test of a single
regime model against the unrestricted model with latitude-dependent regime probabili-
ties. With the unrestricted model, the authors can use the likelihood ratio and simulated
critical values to reject the null hypothesis in favor of the alternative of two regimes.
Because the null distribution derived by Cho and White (2007) applies to only the QLR

constructed using the two-regime model given in (5′) and (6′), we cannot use the QLR

test and, hence, the rscv command to obtain the critical values necessary to evaluate
this unrestricted test statistic.

6 Discussion

We provide a methodology and a new command, rscv, to construct critical values for
a test of regime switching for a simple linear model with Gaussian errors. Despite
the complexity of the underlying methodology, rscv is relatively simple to execute and
merely requires the researcher to provide a range for the standardized distance between
regime means. In section 5, we demonstrate how these methods can be generalized
to a very broad class of models, and we discuss the restrictions necessary to properly
estimate the QLR statistic and use the rscv critical values.

Appendix

The following Stata code was used to create table 2. The code fits the model in section 5
under the alternative hypothesis of two regimes using the EM algorithm and then under
the null hypothesis of one regime using the Stata ml command. Finally, the QLR test
statistic is calculated.

* Estimating QLR test statistic for Bloom, Canning, and Sevilla (2003)

* Log-likelihood function with two regimes
capture program drop llf
program define llf
version 13
args lnf theta1 theta0 delta sigma lambda
quietly replace `lnf´=(1/_N)*((1-etahat)*(ln((2*_pi*`sigma´^2)^(-1/2)) ///

+((-1/(2*`sigma´^2))*(lgdp-`theta0´-`delta´*latitude)^2)+ln(1-`lambda´)) ///
+etahat*(ln((2*_pi*`sigma´^2)^(-1/2))+((-1/(2*`sigma´^2))*(lgdp-`theta1´ ///
-`delta´*latitude)^2)+ln(`lambda´)))

end

* Log-likelihood function for one regime
capture program drop llfsingle
program define llfsingle
version 13
args lnf theta delta sigma
quietly replace `lnf´= (1/_N)*ln(((2*_pi*`sigma´^2)^(-1/2))* ///

exp((-1/(2*`sigma´^2))*(lgdp-`theta´-`delta´*latitude)^2))
end

/***************************************************/
* First, estimate parameters and log likelihood for the case of two regimes:
* lgdp = theta0 + delta*latitude + u~N(0,sigma2) with probability (1-lambda)
* lgpp = theta1 + delta*latitude + u~N(0,sigma2) with probability lambda

/***************************************************/
* Start with initial guess for theta0, theta1, delta, sigma2, and lambda:
regress lgdp latitude
matrix beta=e(b)
svmat double beta, names(matcol)
scalar dhat=betalatitude
generate intercept=lgdp-dhat*latitude
summarize intercept
scalar t0hat=r(mean)-r(Var)



scalar t1hat=r(mean)+r(Var)
scalar shat=sqrt(r(Var))
scalar lhat=0.5
matrix gammahat=(t1hat, t0hat, dhat, shat, lhat)
display "Original guess for parameter values: "
matrix list gammahat

/***************************************************/
* Start loop that continues until parameter estimates have converged
generate error1=10
generate error2=10
generate error3=10
generate tol=1/_N
generate count=0
generate count1=1
generate count2=1
generate count3=1
generate f1=0
generate f0=0
generate fboth=0
generate etahat=0
generate llfhat=0
generate llfnew=0
generate fdelta=0
generate fnew=0
generate Inllfnew=0
generate Inllfdelta=0
generate nd1=0
generate nd2=0
generate nd3=0
generate nd4=0
generate nd5=0

while error1>tol | error2>tol | error3>tol {

* Calculate guess for eta_t=Pr(St=1|sample)
* Calculate f(Yt|St=1, gammahat)
quietly replace f1=((2*_pi*gammahat[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammahat[1,4]^2))*(lgdp-gammahat[1,1]-gammahat[1,3]* ///
latitude)^2)

* Calculate f(Yt|St=0, gammahat)
quietly replace f0=((2*_pi*gammahat[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammahat[1,4]^2))*(lgdp-gammahat[1,2]-gammahat[1,3]* ///
latitude)^2)

* Calculate f(Yt|gammahat)
quietly replace fboth=gammahat[1,5]*f1+(1-gammahat[1,5])*f0
quietly replace etahat=gammahat[1,5]*f1/fboth

/***************************************************/
* Now use etahat to create and maximize log-likelihood function

ml model lf llf /theta1 /theta0 /delta /sigma /lambda
ml init gammahat, copy
ml maximize
matrix gammanew=e(b)

/***************************************************/
* Check whether the parameter estimates have converged
mata: st_matrix("temp", max(abs(st_matrix("gammanew")-st_matrix("gammahat"))))
quietly replace error1=temp[1,1]



* Check whether the log likelihood has converged
quietly replace llfnew=e(ll)
quietly replace llfhat=(1/_N)*((1-etahat) ///

*(ln((2*_pi*gammahat[1,4]^2)^(-1/2)) ///
+((-1/(2*gammahat[1,4]^2)) ///
*(lgdp-gammahat[1,2]-gammahat[1,3]*latitude)^2) ///
+ln(1-gammahat[1,5]))+etahat*(ln((2*_pi*gammahat[1,4]^2)^(-1/2)) ///
+((-1/(2*gammahat[1,4]^2)) ///
*(lgdp-gammahat[1,1]-gammahat[1,3]*latitude)^2) ///
+ln(gammahat[1,5])))

quietly summarize llfhat
quietly replace llfhat=r(sum)
quietly replace error2=abs(llfhat-llfnew)

* Check whether the numeric derivative is zero
* Recalculate incomplete log likelihood with new gamma estimates
quietly replace f1=((2*_pi*gammanew[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,1]-gammanew[1,3]*latitude)^2)
quietly replace f0=((2*_pi*gammanew[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,2]-gammanew[1,3]*latitude)^2)
quietly replace fnew=gammanew[1,5]*f1+(1-gammanew[1,5])*f0
quietly replace Inllfnew=log(fnew)
quietly summarize Inllfnew
quietly replace Inllfnew=r(sum)/_N
* Calculate incomplete log likelihood for gamma + 0.0001
forvalues i=1/5 {

matrix gammadelta=gammanew
matrix gammadelta[1,`i´]=gammadelta[1,`i´]+.0001
quietly replace f1=((2*_pi*gammadelta[1,4]^2)^(-1/2)) ///

*exp((-1/(2*gammadelta[1,4]^2)) ///
*(lgdp-gammadelta[1,1]-gammadelta[1,3]* ///
latitude)^2)

quietly replace f0=((2*_pi*gammadelta[1,4]^2)^(-1/2)) ///
*exp((-1/(2*gammadelta[1,4]^2)) ///
*(lgdp-gammadelta[1,2]-gammadelta[1,3]* ///
latitude)^2)

quietly replace fdelta=gammadelta[1,5]*f1+(1-gammadelta[1,5])*f0
quietly replace Inllfdelta=log(fdelta)
quietly summarize Inllfdelta
quietly replace Inllfdelta=r(sum)/_N
quietly replace nd`i´=abs(Inllfdelta-Inllfnew)/.0001

}
quietly replace error3=max(nd1,nd2,nd3,nd4,nd5)

/***************************************************/
* Keep track of when each convergence criterion is met
quietly replace count1=count1+1 if error1>tol
quietly replace count2=count2+1 if error2>tol
quietly replace count3=count3+1 if error3>tol

* Update gammahat and overall iteration count
matrix gammahat=gammanew
quietly replace count=count+1

* End of loop
}



/***************************************************/
* Calculate final log likelihood for two regimes
quietly replace f1=((2*_pi*gammanew[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,1]-gammanew[1,3]*latitude)^2)
quietly replace f0=((2*_pi*gammanew[1,4]^2)^(-1/2))* ///

exp((-1/(2*gammanew[1,4]^2))*(lgdp-gammanew[1,2]-gammanew[1,3]*latitude)^2)
generate f2reg=gammanew[1,5]*f1+(1-gammanew[1,5])*f0
generate llf2reg=ln(f2reg)
quietly summarize llf2reg
quietly replace llf2reg=r(sum)
* Output final parameter estimates
display "Final estimated parameter values for two regimes: "
matrix list gammanew
display "Final estimated log likelihood for two regimes: " llf2reg
display "Total number of loop iterations: " count
display "Parameter values converged after " count1 " iterations"
display "Log likelihood value converged after " count2 " iterations"
display "Gradient of Log likelihood converged after " count3 " iterations"

/***************************************************/
* Second, estimate parameters and log likelihood for the case of only one regime:

* Maximize log likelihood with only one regime
* lgdp = theta + delta*lat + u~N(0,sigma2)
quietly summarize intercept
matrix gamma0=(r(mean), dhat, .1)
* Maximize to find new estimate of gamma
ml model lf llfsingle /theta /delta /sigma
ml init gamma0, copy
ml maximize
matrix gammasingle=e(b)

*Calculate log likelihood for one regime with estimated gamma
generate llf1reg=ln(((2*_pi*gammasingle[1,3]^2)^(-1/2))* ///

exp((-1/(2*gammasingle[1,3]^2))*(lgdp-gammasingle[1,1]-gammasingle[1,2]* ///
latitude)^2))

quietly summarize llf1reg
quietly replace llf1reg=r(sum)
* Output final parameter estimates
display "Final estimated parameter values for one regime: "
matrix list gammasingle
display "Final estimated log likelihood for one regime: " llf1reg

/***************************************************/
* Finally, calculate QLR test statistic:
generate QLR=2*(llf2reg-llf1reg)
display "Quasi-likelihood-ratio test statistic of one regime: " QLR


